DECUS
PROGRAM LIBRARY

DECUS NO.	8-267
TITLE	DARIC - DATA REDUCTION IN COLUMNS
AUTHOR	J. J. Antal
COMPANY	Army Materials and Mechanics Research Center Watertown, Massachusetts
DATE	June 19, 1970
SOURCE LANGUAGE	PAL III

DARIC - DATA REDUCTION IN COLUMNS

DECUS Program Library Write-up DECUS No. 8-267

ABSTRACT

DARIC is a formatting and computational program which provides for the reduction of 1 to 6 columns of data entered at the ASR keyboard as a function of an incremented independent variable. The format of rows and columns simulates manually-recorded data, offering a simple transition to computer reduction of data where hand-reduction might have been used previously, and simplifies the consolidation of data output by instruments in a variety of formats. Computation is via the Digital floating-point system through a user's data reduction program, the writing rules for which are simplified and standardized by DARIC. Four versions of the program are given by way of overlays which provide a trade-off between data reduction program space and data storage space in the available core memory.

DARIC - DATA REDUCTION IN COLUMNS

John J. Antal
Army Materials & Mechanics Research Center, Watertown, Mass. 02172

I. THE DARIC PROGRAM

The programs having the general name DARIC (DAta Reduction in Columns) were written originally to assist in the reduction of several long columns of data taken simultaneously during an experiment, but outputted by various devices in a variety of formats.

The DARIC program was written to facilitate the entry of these various types of data into the PDP-8 core memory where a program written to reduce one row of data at a time could operate it. The user's data reduction program for one row is referred to as the D.R. Program in this text. Since this type of data reduction is common to many laboratory procedures, the program was expanded to provide the general data reduction formatting system described here.

1. Location of DARIC

DARIC occupies 3 pages of core memory from address Ø2ØØ through address Ø777, and 2/5 of page Ø. It must be used with the Digital Floating-Point System, version D, Digital 8-5-S-D, which occupies core memory from address 4557 through address 7577, reserving the last page of memory for loaders. The user's D. R. Program begins at address 1ØØØ and may extend through as many as nine memory pages. The core memory between DARIC and the floating-point system, locations 1ØØØ through 4556, is divided between the user's D.R. Program and data storage. The data storage section in each form of DARIC is effectively divided into six "columns". Four forms of DARIC are available, each providing a different trade-off between D. R. Program space and data storage space as listed in Table I. Data reduction runs are easily segmented to remove the restriction on the number of data entries which can be accommodated at one time.

Table I. Forms of DARIC

	Pages Available for D.R. Program	Inclusive Addresses for D.R. Program (octal)	Maximum Number of Data Entries per Column (decimal)
DARIC1	1	1ØØØ - 1174	98
DARIC3	3	1ØØØ - 1577	83
DARIC6	6	1ØØØ - 2377	62
DARIC9	9	1ØØØ - 3177	41

2. Data Entry Formats

Data are entered into memory manually at the Teletype ASR keyboard. Two formats for data entry are available: sequential format and parallel format, referred to as S format and P format. With the P format, one datum is entered into each column and one row of entries is completed before moving to the next row. After all data for each column of a particular row are entered, the undefined variable THETA is incremented and printed out automatically and the next row of data is entered manually. This format is very compact and is particularly useful when all of the columnar data is already available in columnar form, or nearly so.

With the S format, a particular column is chosen, one datum is entered manually, THETA is incremented automatically, and the next datum entered manually, and so on until that particular column is full. Other columns are then selected and filled in succession until all data are entered. The S format is particularly useful for entering data which may be available in a variety of formats.

If data is to be entered which does not contain a uniformly incremented variable, THETA may be ignored and the nonuniform variable entered into one of the data columns. When it is desired to obtain the calculation of an expression as a function of one of its variables, the variable may be designated as THETA and DARIC will output a succession of computations for equal increments of the variable.

3. Loading DARIC

Normal use requires that the D.R. Program be written before data is entered. Thus, three programs are in hand when the PDP-8 is to be loaded: Digital 8-5-S-D, DARIC, and the D.R. Program. Each is loaded with the binary loader (S. A. = 7777), but the Floating-Point System, Digital 8-5-S-D, must be loaded before DARIC.

II. THE DIGITAL FLOATING-POINT SYSTEM

The floating-point system of calculation devised for the PDP-8 computer is available in four different versions. Version 4, or D, must be used with DARIC. This version provides for the output of data in both decimal and floating-point decimal notations and provides both exponential and trigonometric operations in addition to the arithmetic operations of the basic floating-point system.

The Interpreter of version D of the floating-point system recognizes the commands in Table II, listed with their mnemonics and octal codes. Unless specifically exempted, the result of these operations is left in the floating accumulator.

TABLE II. FLOATING-POINT SYSTEM COMMANDS

1000	FADD	Add the contents of the effective address to the contents of the floating accumulator.
2000	FSUB	Subtract the contents of the effective address from the contents of the floating accumulator.
3000	FMPY	Multiply the contents of the floating accumulator by the contents of the effective address.
4000	FDIV	Divide the contents of the floating accumulator by the contents of the effective address.
5000	FGET	Get the contents of the effective address and deposit it in the floating accumulator after clearing the floating accumulator.
6000	FPUT	Put the contents of the floating accumulator into the locations specified by the effective address and leave the floating accumulator unchanged.
7000	FNOR	Normalize the contents of the floating accumulator.
0000	FEXT	Exit the floating-point system with the Accumulator cleared.
0001	SQUARE	Square the contents of the floating accumulator.
0002	SQROOT	Take the square root of the contents of the floating accumulator.
0003	FSIN	Take the sine of the angle (in radians) which is in the floating accumulator.
0004	FCOS	Take the cosine of the angle (in radians) which is in the floating accumulator.
0005	ARCTN	Take the arctangent of the contents of the floating accumulator, leaving the result (in radians) in the floating accumulator.
0006	FEXP	Raise e to the power indicated by the contents of the floating accumulator.
0007	FLOG	Take the natural logarithm of the absolute value of the contents of the floating accumulator.
0010	FNEG	Negate the contents of the floating accumulator.
0013	INPUT	Allow input of numbers via Teletype keyboard.

The permanent symbol table of the PAL III assembler contains the mnemonic only for commands 1ØØØ through ØØØØ in the listing above. The remaining commands must be defined for the assembler somewhere in the written data reduction program. The form

FSIN=ØØØ3

is most convenient. FNEG is inserted by DARIC, as it is very useful for data reduction.

Input

Input to the floating-point system is via the Teletype keyboard. Numbers may be entered in decimal notation or floating-point decimal, using the letter E to indicate multiplication by 1Ø raised to the power of the following number. Both the mantissa and the exponent may be signed. For example:

45.773	-.11459
Ø.45773E+2	-Ø.11459
.45773E+2	-11459E-5
45773E-3	-11459E-Ø5

After typing the numerals in a particular entry, a terminator must be typed. Any character which is not used in the input format may be used as a terminator in general, but DARIC requires that only certain terminators be employed in order to control format presentation and to provide for error correction. These are discussed later.

Output

The typed output from Digital 8-5-S-D is either floating decimal or decimal. The format of the output must be specified by the user prior to calling for output. If decimal format is desired, the contents of address ØØ62 must be equal to the total number of digits required, and the contents of the Accumulator must equal the number of digits to the right of the decimal point. If floating decimal output is desired, it is required only that location ØØ62 be cleared.

Note that Digital 8-5-S,D does not contain facilities for rounding off a number to the nearest decimal digit before printing. Thus, in writing the D.R. Program, the user should take care to request a number of digits large enough to indicate the proper magnitude of the nymber. For example, if only one digit to the right of the decimal point was required and so requested, the output of the number 1Ø.1 might read "1Ø.Ø", containing a 1% error. Actually, the number is stored in the floating-point system as 1Ø.Ø9999.

III. WRITING THE DATA REDUCTION PROGRAM

The D. R. Program must be written in a format which connects it to DARIC, but otherwise the program is written as described in the floating-point system manual. An illustration of a D.R. Program is given below with comments on each instruction and notes covering the twelve parts indicated in the left margin. The notes explain the simplified procedures effected by DARIC.

The D.R. Program example repeats a computation on data which had been reduced manually with a desk calculator. Five items of data were recorded in this experiment on neutron scattering from polycrystalline iron. An angular parameter 2θ was incremented uniformly during the course of the experiment, and four parameters representing background intensity and beam intensity ratios were recorded for each value of 2θ. The following column assignments were made by the user:

THETA	Column 1	Column 2	Column 3	Column 4
2θ	R	R_t	B_o	B_T

The cross section S is to be determined from the expression

$$S = -7.4277\ell n\{0.1875R + 0.1875RR_t \times 10^{-6}[0.41667B_o - (2B_T/R)]\}.$$

The constants in the expression contain factors regarding the normalization of count data, and the physical parameters of the iron sample. In addition, 2θis related to the wavelength λ by the expression

$$\lambda = 4.941\sin(2\theta/2).$$

It is desired to reduce the data to values of S and λ for each value of 2θ. Three columns of data are to be output: 2θ (automatically), S, and λ.

The D.R. Program for this computation is given below with the following names assigned by the user to the constants in the expressions for S and λ:

Ø.41667	= CONST1
$Ø.1875 \times 1Ø^{-6}$	= CONST2
Ø.1875	= CONST3
2	= TWO
-7.4277	= MCONST
4.941	= PB2ØØ

NOTE	ADDRESS	INSTRUCTION	COMMENTS
1		FSIN=ØØØ3	
		FLOG=ØØØ7	
2	*1ØØØ		
3		TAD Ø165	/5 digits total in printed output.
		DCA ØØ62	

NOTE	ADDRESS	INSTRUCTION	COMMENTS
4		JMS I 77	/Enter data via DARIC.
		JMS I 7	/Call F. P. Interpreter
		FGET I 74	/Get B_T from Column 4.
		FDIV I 71	/Divide by R of Column 1.
		FMPY TWO	/Multiply by 2.
		FPUT I 76	/Store result temporarily in Column 6
		FGET I 73	/Get B_o from Column 6.
		FMPY CONST1	/Multiply B_o by Ø.41667.
		FSUB I 76	/Subtract term stored in Column 6.
		FMPY CONST2	/Multiply by Ø.1875 x $1Ø^{-6}$.
5		FMPY I 71	/Multiply by R from Column 1.
		FMPY I 72	/Multiply by R_t from Column 2.
		FPUT I 76	/Store result temporarily in Column 6.
		FGET CONST3	/Get Ø.1875.
		FMPY I 71	/Multiply by R of Column 1.
		FADD I 76	/Add the stored term.
		FLOG	/Take the natural log.
		FMPY MCONST	/Multiply by -7.4277.
		FEXT	/Leave the F.P. System
		TAD Ø163	/3 digits to right of d.p.desired.
6		JMS I 6	/Print out result
		TAD MFIVE	/Deposit a -5 at address
7		DCA Ø175	/Ø175 to provide a
		JMS I 64	/Tab of 4 spaces
		JMS I 7	/Call F.P. Interpreter.
		FGET I 167	/Get THETA of this row.
		FDIV TWO	/Divide by 2.
8		FMPY RADEG	/Convert degrees to radians.
		FSIN	/Take the sine.
		FMPY PB2ØØ	/Multiply by 4.941.
		FEXT	/Leave the F.P. System.
		TAD Ø164	/4 digits to right of d.p.desired.
9		JMS I 6	/Print out A.
10		JMP I 66	/Jump to DARIC for next row of data.
	CONST1,	7777	/Ø.41667
		3252	
		5343	
	TWO,	ØØØ2	/2
		2ØØØ	
		ØØØØ	
	CONST2,	7752	/Ø.1875E-6
		3112	
		4716	
	CONST3,	7776	/Ø.1875
		3ØØØ	
11		ØØØØ	
	MCONST,	ØØØ3	/-7.4277
		4222	
		4Ø43	
	RADEG,	7773	/Ø.Ø174533 Rad/Deg

NOTE	ADDRESS	INSTRUCTION	COMMENTS
		2167	
		6435	
	PB2ØØ	ØØØ3	/4.941
		236Ø	
		7126	
	MFIVE,	-5	/-5 (Not F.P.)
12	$		

NOTE 1: Those mnemonics which are not a part of the Assembler's permanent symbol table must be defined for the Assembler. This position in the program is convenient. The loading of registers ØØØ6 and ØØØ7 is accomplished by DARIC and need not be repeated here as shown in the floating-point system manual. The Teletype buffers are also cleared by DARIC and initial instructions such as KCC and TLS are not required in the D.R. Program.

NOTE 2: In order to make full use of the D.R. Program space provided, the starting address of the D.R. Program should always be 1ØØØ.

NOTE 3: DARIC provides for the output of printed data to a total of 6 digits in each number. If, as in this example, this is not satisfactory, the maximum number of digits may be altered by depositing that number in register ØØ62. If register ØØ62 is cleared, all numbers printed out will appear in floating-point decimal form with seven significant figures.

Any other initialization required to be accomplished only once each time the program is started at address 1ØØØ must be included in the program before the instruction JMS I 77 is listed.

NOTE 4: JMS I 77 must be the instruction just prior to the start of computation. This sends the program to a subroutine in DARIC (the major portion of DARIC) which formats the input of data into core memory. The entry of data is discussed later.

NOTE 5: This is the computation of S for one ROW of data. Note that access to the experimental data which has been stored in columns 1 through 6 is obtained by indirect address to locations 71 through 76, respectively. Note also that temporary floating-point storage is easily obtained by depositing in columns which do not contain data.

NOTE 6: These instructions provide for the printout of data. If the floating decimal output had been chosen (see NOTE 3), only the instruction JMS I 6 would have been necessary. If no digits were to appear to the right of the decimal point in this decimal output, only the instruction JMS I 6 would have been necessary since the Accumulator would have been cleared in the previous instruction, FEXT. If, as in this example, the decimal format is desired with several digits to the right of the decimal point, the number of digits must be put into the Accumulator just prior to calling JMS I 6.

A series of numbers have been located on page Ø of core memory by DARIC which may be used to select the output decimal position. One of the following listed instructions amy be used to produce the listed results when JMS I 6 is called:

Instruction	Number of digits to right of decimal point
CLA	Ø, no decimal point printed
IAC	1
TAD Ø162	2
TAD Ø163	3
TAD Ø164	4
TAD Ø165	5
TAD Ø166	6

The numbers stored in registers Ø162 through Ø166 may be used for other purposes, such as filling register ØØ62 (see NOTE 3), but they must not be altered.

NOTE 7: The result of the computation of S having been printed out, it is desired in this example to compute λ and output the result in an adjoining column. DARIC contains a tabbing routine which may be used for this purpose. It is called with the instruction JMS I 64. The tabbing routine puts out ten successive space instructions to the Teletype. If the space desired between outputted columns is less than ten spaces, the negative of (number of spaces + 1) must be placed in register Ø175 just prior to calling the instruction JMS I 64. The example shown here produces four spaces after the value of S is printed.

NOTE 8: This is the computation of λ from the given expression which is a function of 0. DARIC provides access to the current value of THETA in floating-point 3-word notation by indirect reference to location Ø167. In the sole case of DARIC1, THETA is found in the last three registers of the D.R. Program memory page (locations 1175, 1176, and 1177) where it may be addressed directly. Through address Ø167, the value of THETA may be included in a floating-point program at any time.

NOTE 9: λ is here printed out with four digits to the right of the decimal place.

NOTE 10: The last instruction of the D.R. Program must be JMP I 66. This instruction sends the program to DARIC for printout of the next value of THETA and an advance to the next row of data. DARIC initiates the repetition of the data reduction computation on this new row of data by returning to the instruction immediately following JMS I 77.

NOTE 11: As is normal with programming, all constants and storage registers which must be used for the computation are listed at the end of the program. Note that in this example 7 constants were listed in floating-point notation. The 3-word constants may be formulated as described in the previous chapter, but the floating-point system itself may be used to produce the 3 words by assembling the following program:

```
/    FPCON                    DECIMAL CONSTANTS TO F.P. CONSTANTS - REQUIRES
/                             DIGITAL 8-5-S-A, FPNTA, TERMINATE DECIMAL
/                             ENTRY AT ASR WITH A LINE FEED. S.A. = Ø2ØØ.
```

```
*0005
          7400
*0200
          KCC
          TLS
INPUT,    JMS I 5
          TAD 44
          JMS PRINT
          TAD 45
          JMS PRINT
          TAD 46
          JMS PRINT
          TAD CR
          JMS I OUT
          TAD LF
          JMS I OUT
          JMP INPUT

PRINT,    0000
          DCA POT
          TAD TUFIVE
          JMS I OUT
          TAD TUFIVE
          JMS I OUT
          TAD POT
          RTL
          RTL
          JMS TYPE
          TAD POT
          RTR
          RTR
          RTR
          JMS TYPE
          TAD POT
          RTR
          RAR
          JMS TYPE
          TAD POT
          JMS TYPE
          JMP I PRINT

TYPE,     0000
          AND BLOT
          JMS I OUTDG
          JMP I TYPE

CR,       215
LF,       212
BLOT,     0007
OUT,      7344
OUTDG,    7352
POT,      0
TUFIVE,   255
$
```

NOTE 12: As always, the program is terminated with a dollar sign.

IV. DATA ENTRY AND COMPUTATION

1. P and S Formats

Two formats for the entry of data are available in each DARIC program: parallel and sequential (P and S). P format allows the entry of data into all columns one row at a time, S format allows the entry of all data into a particular column before any other column is entered. The coice of format and initial information regarding the parameter THETA are accumulated by DARIC in a dialog with the user at the Teletype keyboard.

After the F.P., DARIC, and D.R. Programs have been loaded, the computer is started at address 1ØØØ. Reproduced below is the dialog format for the example program of Chapter III. The portions shown underlind were entered by the user. The symbol ¢ indicates a carrier return, and # indicates a space.

```
FIRST THETA 1Ø7¢                    COLUMNS IN P OR S? P
DELTA THETA .5Ø¢
LAST THETA  +155.49

THETA              1          2          3          4          5          6

+1Ø7.ØØ          .523#      810#       39#        16¢
```

When the S format is requested, the dialog is as follows:

```
FIRST THETA 1Ø7¢                    COLUMNS IN P OR S? S
DELTA THETA .5Ø¢
LAST THETA  +155.49

COLUMN NO. 1

+1Ø7.ØØ     .523¢
+1Ø7.49
```

FIRST THETA is the initial value of the variable and DELTA THETA is the desired uniform increment of THETA. Since the values of FIRST THETA and DELTA THETA are entered via the floating-point input, they may be entered in any of the forms listed earlier and must be terminated. A carrier return should be used as the terminator in each case to preserve the format. DELTA THETA should be entered with the full number of digits to the right of the decimal point as will be desired in the printout of THETA, for DARIC uses this DELTA THETA value to make the determination.

If none of the parameters in the data under consideration is uniformly incremented, it is convenient to set FIRST THETA = 1 and DELTA THETA = 1. The column headed THETA will then number the rows of data. The nonuniformly incremented parameter is entered into one of the six columns.

The LAST THETA value is the largest value of the parameter THETA which can be accommodated in core memory by the particular version of DARIC being used.

In the P format, column headings are printed out by DARIC, followed by the value of the FIRST THETA at the left margin. At this point, the keyboard is ready for input of data by the user. Again, the data enters through the floating-point system input and each number must be terminated. In P format, the space bar is depressed for the normal terminator. However, *the terminator of the last entry in the first row must be a carrier return.* After this carrier return, DARIC lists the next value of THETA at the left margin and waits for the next series of data entries to be made.

In the S format, a request for a COLUMN NUMBER is made. The only proper replies are the numbers 1 through 6. No terminator is required, and columns may be selected in any order. The value of the FIRST THETA is printed out by that column, terminating the number with a carrier return. Successive THETAs are printed by DARIC and each datum entered should be terminated by a carrier return. In both formats, L must be the terminator at the end of the last row of data.

2. Terminators and Errors

The standard error-correction device provided by Digital 8-5-S is always operable when using DARIC. Typing a RUBOUT after an error has been made in a particular digit will make the floating-point input ignore all digits entered and the number may be begun again immediately after the RUBOUT.

As data is entered in each column, four different terminators may be used. Their use is summarized below:

CARRIER RETURN... Normal terminator for all entries in S format.
Terminator for last entry of first row in P format.
Line feed follows this terminator without operator input.

SPACE..... Normal terminator for all entires in P format except last entry in first row.

L..... L(ast entry). Must be the terminator for the last entry of the first column filled in S format.
Must be the terminator of the very last entry in P format.
Must be used, but only once, each time program is started.
L does not print out.

M..... M(istake). May be used to terminate an entry at any time; all previously entered data for that row is ignored and the value of THETA is repeated, ready for entry of the proper data. M prints out to provide error notation.

3. No Entry

Occasionally an experimenter will delete data points obtained in the course of an experiment because they are in error due to obvious causes. If these deletions are made in only one of several parameters recorded in a particular experiment, then it would be necessary to note the particular values of THETA at which they occur since a data reduction computation would be invalid at those points. To facilitate this process, DARIC allows the entry of the letter N (No Entry) in the position of the deleted datum. N is terminated in the same manner as a number. The entry of N places a very large number in core memory which usually results in the appearance of zero, XXXXXX (a number beyond the capacity of the F.P. system), a blank space, or another number out of context with those on either side of it. As a result, the positions of the omitted data points become clear in the tabulated results of the computation without having otherwise recorded the fact of omission.

4. The D. R. Computation

After all data has been entered at the Teletype keyboard, DARIC will request:

COLUMN NO.

The data reduction computation begins immediately after the operator replies with a zero, Ø. While the program is stopped, awaiting a reply to this request, the operator has an opportunity to check the entered data before the computation proceeds, or he may wish to prepare column headings for the outputted results by taking the Teletype off-line. Column headings may be entered on the line below the "COLUMN NO." request and the platen manually returned to its original position.

When the results of the computation are printed out for all values of THETA for which data was available, the computer stops.

5. Repeat of Computation

If, after completion of the output of the results of the computation, it is desired to repeat the calculation, this is accomplished by depressing the "CONT" key on the computer console and answering the request for "COLUMN NO." with a Ø.

6. Alteration of Data

If, after completion of the output of the results of the computation, it is desired to change the data in one or more of the columns and then repeat the calculation, depress the "CONT" key on the PDP-8 console. The COLUMN NO. request will be made, and the operator is free to request any column from 1 through 6. Data may now be entered as if the user were employing the S format. When the total number of entries equals that of the

previously computed data, the computer will halt. Depress the "CONT" key again and answer the COLUMN NO. request with any number Ø through 6 to obtain a computation involving the new data or to alter an additional column of data.

The use of this system of data alteration requires that data be placed in a column for all values of THETA used previously. Also, the terminator L must not be used after the last entry since the program is not being restarted at address 1ØØØ.

7. Computation of Theoretical Expressions

DARIC may be applied to computations which are wholly or partially derived from theory. The computational program is written exactly as a D.R. Program for one row of data, except for the additional necessity to instruct DARIC on the number of computations to be made (this is determined by the terminator L during data reduction). This instruction of DARIC is accomplished by depositing an octal number, χ, in locations Ø371 and Ø777, where

$$\chi = 1 - \text{Number of Computations Desired},$$

prior to the instruction JMS I 77.

Any number of computations up to the full data entry capacity of DARIC, listed in Table I, can be obtained on each pass. If the number of computations desired equals the full data entry capacity of DARIC, the initial instructions loading χ may be deleted, as DARIC will always stop when the LAST THETA value is reached.

V. PROGRAM LISTINGS

```
/   DARIC 1          1 PAGE OF MEMORY FOR D.R. PROGRAM
/                    98 DATA ENTRIES PER COLUMN
/   PART 1           REQUIRES FPNTD, DIGITAL 8-5-S,D(4/18/65)

*0006
          7200       /OUTPUT LOCATION
          5600       /INTERPRETER LOCATION
*0062
          0006       /NO. OF DIGITS IN OUTPUT
          0000       /NO. OF DIGITS RT. OF DECIMAL PT. IN THETA
          0420       /TAB LOCATION
          0000       /CURRENT COLUMN NUMBER
          0662       /INCALL, RETURN FROM D.R. PROGRAM
          0000       /RETURN TO D.R. PROGRAM
          7777       /P,S SWITCH
          1211       /COLUMN 1
          1657       /COLUMN 2
          2325       /COLUMN 3
          2773       /COLUMN 4
          3441       /COLUMN 5
          4107       /COLUMN 6
          0244       /S. A. OD DARIC
*0161
          7777       /COLUMN 0 SWITCH
             2       /DECIMAL POSITION NUMBERS
             3
             4
             5
             6
          1175       /THETA ACCUMULATOR F. P. LOCATION
          1200       /FIRST THETA F.P. LOCATION
          1203       /DELTA THETA F.P. LOCATION
          1206       /MAX. NO. LINES, F.P. LOCATION
          0000       /N SWITCH
          0000       /DELTA THETA SWITCH
           -13       /NEG. OF NO. SPACES+1
          0000       /ROW COUNTER
*0371
          -141       /NOLINS

*0771
          1211       /FIR1
          1657       /FIR2
          2325       /FIR3
          2773       /FIR4
          3441       /FIR5
          4106       /FIR6
          -141       /NOLIN
```

```
                                N
*1206
          0007          /F.P.141 (OCTAL)
          3020
          0000

PAUSE


/    DARIC3             3 PAGES OF MEMORY FOR D.R. PROGRAM
/                       83 DATA ENTRIES PER COLUMN
/    PART 1            REQUIRES FPNTD, DIGITAL 8-5-S,D  4/18/65

*0006
          7200          /OUTPUT LOCATION
          5600          /INTERPRETER LOCATION
*0062
          0006          /NO. OF DIGITS IN OUTPUT
          0000          /NO. OF DIGITS RT. OF DEC. PT. IN THETA
          0420          /TAB LOCATION
          0000          /CURRENT COLUMN NUMBER
          0662          /INCALL, RETURN FROM D.R. PROGRAM
          0000          /RETURN TO D.R. PROGRAM
          7777          /P,S SWITCH
          1614          /COLUMN 1
          2205          /COLUMN 2
          2576          /COLUMN 3
          3167          /COLUMN 4
          3560          /COLUMN 5
          4151          /COLUMN 6
          0244          /S.A. OF DARIC
*0161
          7777          /COLUMN 0 SWITCH
             2          /DECIMAL POSITION NUMBERS
             3
             4
             5
             6
          1600          /THETA ACCUMULATOR F.P. LOCATION
          1603          /FIRST THETA F.P. LOCATION
          1606          /DELTA THETA F.P. LOCATION
          1611          /MAX. NO. LINES, F.P. LOCATION
          0000          /N SWITCH
          0000          /DELTA THETA SWITCH
           -13          /NEG. OF NO. SPACES+1
          0000          /ROW COUNTER

*0371
          -122          /NOLINS
```

```
*0771
          1614        /FIR1
          2205        /FIR2
          2576        /FIR3
          3167        /FIR4
          3560        /FIR5
          4151        /FIR6
          -122        /NOLIN
*1611
          0007
          2440
          0000

PAUSE


/   DARIC6            6 PAGES OF MEMORY FOR D.R. PROGRAM
/                     62 DATA ENTRIES PER COLUMN
/   PART 1           REQUIRES FPNTD, DIGITAL 8-5-S,D  4/18/65

*0006
          7200        /OUTPUT LOCATION
          5600        /INTERPRETER LOCATION
*0062
          0006        /NO. OF DIGITS IN OUTPUT
          0000        /NO. OF DIGITS RT. OF DEC. PT. IN THETA
          0420        /TAB LOCATION
          0000        /CURRENT COLUMN NUMBER
          0662        /INCALL, RETURN FROM D.R. PROGRAM
          0000        /RETURN TO D.R. PROGRAM
          7777        /P,S SWITCH
          2414        /COLUMN 1
          2707        /COLUMN 2
          3202        /COLUMN 3
          3475        /COLUMN 4
          3770        /COLUMN 5
          4263        /COLUMN 6
          0244        /S.A. OF DARIC
*0161
          7777        /COLUMN 0 SWITCH
             2        /DECIMAL POSITION NUMBERS
             3
             4
             5
             6
          2400        /THETA ACCUMULATOR F.P. LOCATION
          2403        /FIRST THETA F.P. LOCATION
          2406        /DELTA THETA F.P. LOCATION
          2411        /MAX. NO. ROWS, F.P. LOCATION
```

```
          0000        /N SWITCH
          0000        /DELTA THETA SWITCH
           -13        /NEG. OF NO. SPACES+1
          0000        /ROW COUNTER
*0371
          -75         /NOLINS
*0771
          2414        /FIR1
          2707        /FIR2
          3202        /FIR3
          3475        /FIR4
          3770        /FIR5
          4263        /FIR6
           -75        /NOLIN
*2411
          0006
          3640
          0000

PAUSE


 /   DARIC9            9 PAGES OF MEMORY FOR D.R. PROGRAM
/                     41 ENTRIES PER COLUMN
/   PART 1           REQUIRES FPNTD, DIGITAL 8-5-S,D  4/18/65

*0006
          7200        /OUTPUT LOCATION
          5600        /INTERPRETER LOCATION
*0062
          0006        /NO. OF DIGITS IN OUTPUT
          0000        /NO. OF DIGITS RT. OF DEC. PT. IN THETA
          0420        /TAB LOCATION
          0000        /CURRENT COLUMN NUMBER
          0662        /INCALL, RETURN FROM D.R. PROGRAM
          0000        /RETURN TO D.R. PROGRAM
          7777        /P,S SWITCH
          3214        /COLUMN 1
          3407        /COLUMN 2
          3602        /COLUMN 3
          3775        /COLUMN 4
          4170        /COLUMN 5
          4363        /COLUMN 6
          0244        /S.A. OF DARIC
*0161
          7777        /COLUMN 0 SWITCH
             2        /DECIMAL POSITION NUMBERS
             3
             4
             5
             6
          3200        /THETA ACCUMULATOR F.P. LOCATION
```

```
          3203        /FIRST THETA F.P. LOCATION
          3206        /DELTA THETA F.P. LOCATION
          3211        /MAX. NO. ROWS F.P. LOCATION
          0000        /N SWITCH
          0000        /DELTA THETA SWITCH
           -13        /NEG. OF NO. SPACES+1
          0000        /ROW COUNTER
*0371
           -50        /NOLINS
*0771
          3214        /FIR1
          3407        /FIR2
          3602        /FIR3
          3775        /FIR4
          4170        /FIR5
          4363        /FIR6
           -50        /NOLIN
*3211
          0006
          2400
          0000

PAUSE


/   DARIC             PART 2

*0177
MESAGE,   0000        /DIGITAL 8-18-U WITH ALTERATIONS
          CLA CMA
          TAD MESAGE
          DCA 10
          TAD I 10
          DCA MSRGHT
          TAD MSRGHT
          RTR
          RTR
          RTR
          JMS TYPECH
          TAD MSRGHT
          JMS TYPECH
          JMP MESAGE+4
MSRGHT,   0
TYPECH,   0
          AND MASK77
          SNA
          JMP I 10
          TAD M40
          SMA
          JMP.+3
          TAD C340
          JMP MTP
```

```
          TAD M3
          SZA
          JMP.+3
          TAD LF
          JMP MTP
          TAD M2
          SZA
          JMP .+3
          TAD CR
          JMP MTP
          TAD C245
MTP,      JMS I OUT
          JMP I TYPECH

          INPUT=0013
START,    0
          TAD START
          DCA 0067
          KCC
          TLS
          JMS 0177
                              4543; 0000
          TAD M40
          DCA 0175
          JMS I 0064
          JMS 0177
          0317; 1425; 1516; 2340  /COLUMNS
          1116; 4020; 4017; 2240  /IN P OR
          2377; 4000              /S?
          JMS I TYPE
          TAD MPEE
          DCA 0070                /P,S SWITCH SET. 0=P
          DCA 0063                /RESET THETA DIGITS
          JMS 0177
                             4545; 0611; 2223; 2400/FIRST
          JMS THETA
          JMS I 7
          INPUT
          FPUT I 0167
          FPUT I 0170
          FEXT
          JMS 0177
          0405; 1424; 0100        /DELTA
          JMS THETA
          STA                    /SET DELTA THETA SWITCH
          DCA 0174
          JMS I 7
          INPUT
          FPUT I 0171
          FEXT
          DCA 0174               /RESET DELTA THETA SWITCH
          JMS 0177
```

```
          7615; 0130; 5600        />MAX.
          JMS THETA
          JMS I 7
          FGET I 0172
          FMPY I 0171
          FADD I 0170
          FEXT
          TAD 0063
          JMS I 6
          JMS 0177
          4543; 4300
          TAD NOLINS              /SET MAX. NO. OF LINES
          DCA 0176
          DCA I ZEND
          STA
          DCA 0161               /RESET COLUMN SWITCH
          TAD 0070
          JMP I P2

THETA,    0000
          JMS 0177
                                 4024; 1005
          2401; 4000
          JMP I THETA

ZEND,     0756
OUT,      7344
CR,       215
MASK77,   77
M40,      -40
C340,     340
M3,       -3
NOLINS,   -141
LF,       212
M2,       -2
C245,     245
MPEE,     -320
TYPE,     7142
P2,       0400
PAUSE


/         DARIC                   PART 3

*0400
          SZA
          JMP REQUST              /S(EQUENTIAL) ENTRY
          JMS I THET              /P(ARALLEL) ENTRY
          TAD MEIGHT              /TYPE COLUMN HEADINGS
          DCA 0175
          JMS SPACES
          TAD TUSIX1
```

```
          DCA MAPLE
          JMS HEAD
          JMS HEAD
          JMS HEAD
          JMS HEAD
          JMS HEAD
          JMS HEAD
PARLEL,   TAD TUSIX1
          JMP COLUMN              /CHOOSE COLUMN 1

SPACES,   0000
          ISZ 0175                /TEN SPACES PER COLUMN
          JMP .+4
          TAD MWUN3
          DCA 0175
          JMP I SPACES
          TAD SP
          JMS I OUTS
          JMP SPACES+1

HEAD,     0000
          TAD MAPLE               /HEADINGS GENERATOR
          JMS I OUTS
          ISZ 0175

          ISZ MAPLE             JMS SPACES
          JMP I HEAD

REQUST,   JMS 0177
          0317; 1425; 1516        /REQUEST FOR COLUMN NUMBER
          4016; 1756; 4000
          JMS I TYP
COLUMN,   DCA 0065
          JMS 0177
          4543; 4300
PARCOL,   TAD 0065
          CIA
          TAD TUSIX1
          SPA SNA
          JMP DATAIN
          CLA                     /COLUMNS 1-6
          DCA 0161                /COLUMN 0
          STA
          DCA 0070              /USE S FORMAT FOR COLUMN 0
          JMS LINE
          JMP I 0067
DATAIN,   SMA
          JMP COL1
          IAC
          SMA
          JMP COL2
```

```
            IAC
            SMA
            JMP COL3
            IAC
            SMA
            JMP COL4
            IAC
            SMA
            JMP COL5
            CLA
            NOP
            TAD 0076; JMP ADDR
COL5,       TAD 0075; JMP ADDR
COL4,       TAD 0074; JMP ADDR
COL3,       TAD 0073; JMP ADDR
COL2,       TAD 0072; JMP ADDR
COL1,       TAD 0071                    /DATA ENTRY ADDRESS HELD
ADDR,       DCA HOLD
            TAD 0070
            SZA CLA
            JMP ON
            TAD 0065
            CIA
            TAD TUSIX1
            SNA CLA                     /TYPE OUT THETA
ON,         JMS LINE
            JMS I 7
            INPUT
            FPUT I HOLD                 /DATA ENTRY STORED
            FEXT
            ISZ 0173                    /N TYPED?
            JMP SPT
            JMS I 7
            FGET MANT
            FPUT I HOLD
            FEXT                        /RESET N SWITCH
            DCA 0173                    /SPACE TYPED?
SPT,        TAD 0057
            CIA
            TAD SP
            JMP I EST

LINE,       0                           /TYPE THETA ROUTINE
            TAD MFIVE
            DCA 0175
            JMS I 7
            FGET I 0167
            FEXT
            TAD 0063
            JMS I 6
            JMS SPACES
            JMP I LINE
```

```
MFIVE,      -5
HOLD,       0
MAPLE,      0
THET,       0354
MANT,       2000
TUSIX1,     261
SP,         240
EST,        6573
TYP,        7142
OUTS,       7344
MWUN3,      -13
MEIGHT,     -10
PAUSE


/           DARIC                       PART 4

*0600
            TAD EMMSP                   /M TYPED?
            SZA CLA
            JMP LENTRY                  /NOT M
            JMP CRLF                    /M
ENDT,       TAD 0070
            SZA CLA
            JMP INCALL                  /S
            TAD END                     /P
            CIA
            TAD 0065
            SNA CLA                     /END COLUMN?
            JMP INCALL                  /YES, TYPE CR, LF
            JMS I 0064                  /NO, FILL COLUMN WITH SPACES
            ISZ 0065                    /P
            JMP I PARCO
MOVALL,     TAD 0071; TAD 0163          /INCREMENT F.P. ADDRESSES
            DCA 0071
            TAD 0072; TAD 0163
            DCA 0072
            TAD 0073; TAD 0163
            DCA 0073
            TAD 0074; TAD 0163
            DCA 0074
            TAD 0075; TAD 0163
            DCA 0075
            TAD 0076; TAD 0163
            DCA 0076
            JMS I 7                     /INCREMENT THETA
            FGET I 0167
            FADD I 0171
            FPUT I 0167
            FEXT
```

```
            ISZ 0176                /INCREMENT LINE NO.
            NOP
            JMP SAMLIN
CRLF,       JMS 0177
            4543; 0000
SAMLIN,     TAD 0070
            SZA CLA
            JMP I PARCO
            TAD TUSIX
            DCA 0065
            JMP I PARCO
INCALL,     JMS 0177                /RETURN FROM D.R. PROGRAM
            4543
            0000
LENTRY,     TAD 0176                /LAST LINE?
            SZA CLA
            JMP PORS                /NO
            TAD LLINE               /YES, RESET LAST LINE SWITCH
            CIA
            TAD NOLIN
            DCA 0176
            DCA END
            JMS I 7                 /RESET THETA ACCUMULATOR
            FGET I 0170
            FPUT I 0167
            FEXT
            TAD FIR1; DCA 0071      /RESET ALL COLUMN ADDRESSES
            TAD FIR2; DCA 0072
            TAD FIR3; DCA 0073
            TAD FIR4; DCA 0074
            TAD FIR5; DCA 0075
            TAD FIR6; DCA 0076
            JMS 0177
            4343
            0000
            TAD 0161                /COLUMN 0?
            SNA
            HLT                     /YES
            JMP I REQUS             /REQUEST NEXT COLUMN
PORS,       TAD 0070
            SZA CLA
            JMP MOVALL
            TAD 0065
            DCA END
            JMP MOVALL

ENN,        TAD MENN                /N(O ENTRY)?
            SZA CLA
            JMP ELL                 /NOT N
            STA
            DCA 0173
```

```
            TAD 0057
            JMS I OU                /TYPE N
            DCA 0057
            JMP DOWN
ELL,        TAD 0057                /TEST FOR L(AST ENTRY)
            TAD MELL
           SZA CLA
            JMP DOWN
            TAD 0176
            DCA LLINE               /SAVE LINE NUMBER
            DCA 0176                /SET LINE NO. = 0
CHANGE,     TAD CRS
            DCA 0057
DOWN,       TAD 0057
            JMP I ZERO

END,        0
LLINE,      0
CRS,        215
TUSIX,      261
ZERO,       7151
MELL,       -314
REQUS,      0440
MENN,       -316
EMMSP,      315-240
PARCO,      0454
OU,         7344
FIR1,       1211
FIR2,       1657
FIR3,       2325
FIR4,       2773
FIR5,       3441
FIR6,       4107
NOLIN,      -141
PAUSE


/   DARIC               PART 5

*6554
            6000
*6573
            SNA
            JMP I TEND
            JMP I EM
TEND,       0604
EM,         0600

*7040
            JMP I OAK
```

```
*7150
          JMP PINE
*7172
PINE,     ISZ 0175
          NOP
          TAD 0057
          JMP I EN
EN,       0732
OAK,      6173

*6173
          CLA
          TAD I PRSW          /TEST FOR DECIMAL PT.
          JMP I BEECH
BEECH,    6763
PRSW,     7513

*6763
          CMA
          AND 0174            /TEST FOR DELTA THETA
          JMP I ELM
ELM,      4774

*4774
          SZA CLA
          ISZ 0063
          JMP I ASH
ASH,      7017

$
```